# Éléments de correction - Feuille de travaux pratiques # 1

## 2.2 Simulation par fonction de répartition inverse

**Proposition 1** *Soit $X$ une variable aléatoire réelle de fonction de répartition $F$. Pour $u \in [0,1]$, on désigne par $F^{-1}(u) := \inf\{x \in \mathbb{R},\ F(x) \geq u\}$ l'inverse généralisée de la fonction de répartition $F$. Si $U \sim \mathcal{U}_{[0,1]}$ alors $F^{-1}(U)$ a pour fonction de répartition $F$.*

1. À partir de variables uniformes sur $[0,1]$, générer un $n-$échantillon de variables de Bernoulli de paramètre $1/2$ à valeurs dans $E = \{0,1\}$. Même question lorsque $E = \{-1,1\}$.
   U=rand(1,n) ; on genere un vecteur $U$ de longueur $n$ de variables i.i.d. de loi unifome
   Y=bool2s(U<0.5) ; les entrées de $Y$ sont alors i.i.d. de loi unifome sur $\{0,1\}$
   Z=2*Y-1 ; les entrées de $Z$ sont alors i.i.d. de loi unifome sur $\{-1,1\}$
2. Simuler une variable aléatoire $Y$ de loi uniforme à valeurs dans $\{0,1,2,3\}$.
   Soit $U$ de loi unifome sur $[0,1]$, on renvoie alors 0, 1, 2, 3 selon que $U \in [0,1/4]$, $U \in [1/4,1/2]$, $U \in [1/2,3/4]$, ou $U \in [3/4,1]$.
3. À partir de variables uniformes, simuler un $n-$échantillon $(X_1,\ldots,X_n)$ de variables aléatoires de loi exponentielle de paramètre 2. Mettre en évidence la loi des grands nombres en traçant la fonction $k \mapsto (X_1+\ldots+X_k)/k$ pour $k$ allant de 1 à $n$.
   On considère $U = rand(1,n)$ un vecteur de variables i.i.d. de loi unifome. L'inverse de la fonction de répartion $F(x) = 1-e^{-2x}$ est $F^{-1}(y) = -\frac{1}{2}\log(1-y)$ de sorte que $X = F^{-1}(U)$ est un vecteur de variables i.i.d. de loi exponentielle. On pose alors $Z = cumsum(X)$ et l'on illustre la loi des grands nombres en traçant $plot(Z./[1:1:n])$

## 2.3 Simulation par rejet

1. Montrer que si $\varepsilon$ suit une loi de Bernoulli de paramètre $1/2$ dans $\{-1,1\}$ et si $Z$ est une variable exponentielle de paramètre 1 indépendante de $\varepsilon$, alors $\varepsilon Z$ a pour densité la fonction $g$ sur $\mathbb{R}$ ;
   Soit $f$ une fonction continue bornée, on a alors
   $$\mathbb{E}[f(\varepsilon Z)] = \frac{1}{2}\int_{\mathbb{R}^+} f(x)e^{-x}dx + \frac{1}{2}\int_{\mathbb{R}^+} f(-x)e^{-x}dx = \frac{1}{2}\int_{\mathbb{R}} f(x)e^{-|x|}dx,$$
   d'où le résultat.
2. Implémenter un algorithme qui simule une variable gaussienne via la méthode de rejet ;
   Le point précédent permet de générer facilement des variables de densité $g$. Il ne reste plus qu'à appliquer l'algorithme de rejet...

### 2.4 Méthode de Monte-Carlo

Application : Calculer les approximations des intégrales suivantes.

$$I_1 = \int_0^1 4\sqrt{1-x^2}dx \quad \text{(aire du disque unité)},$$

```
// Methode d'estimation de Monte Carlo pour I_1
// avec conservation des intervalles de confiance pour toutes
// les valeurs entre 1 et M
// Cette fonction a vocation a etre utilisee pour visualiser
// graphiquement la stabilsation vers la moyenne
function res=IC_I_1(M)
  Alea=4*sqrt(1-rand(M,1).^2);
  X=cumsum(Alea);
  V=cumsum(Alea.^2);
  Abscisse=[1:M]';
  Moy=X./Abscisse;
  Factor=Abscisse./(max(Abscisse-1,1));
    V=V./Abscisse-Factor.*Moy.^2;
  Dev=1.96*sqrt(V)./sqrt(Abscisse);
  res=[Moy-Dev,Moy,Moy+Dev ];
endfunction

// On fait la meme en version breve, i.e. estimation de la moyenne empirique
// et de l'intervalle à 95%
// associé pour la valeur M en argument
function res=IC_I_1_Short(M)
  Alea=4*sqrt(1-rand(M,1).^2);
  X=sum(Alea);
  V=sum(Alea.^2);
  Moy=X/M;
  Factor=M/(M-1);
    V=V/M-Factor*Moy^2;
  Dev=1.96*sqrt(V)/sqrt(M);
  res=[Moy-Dev,Moy,Moy+Dev ];
endfunction
```

$$I_3 = \int_{[-1,1]^3} \mathbb{1}_{\{x^2+y^2+z^2\leq 1\}} dxdydz, \quad \text{(volume de la boule unité)}$$

```
// Version brève pour l'estimation de I_3
function res=IC_I_3_Short(M)
  Alea=((-1+2*rand(M,3)).^2*ones(3,1)<=1);
  X=sum(Alea);
  V=sum(Alea); // Comme le retour aleatoire est 0 ou 1 on ne change pas en passant au carre
  Moy=X/M;
  Factor=M/(M-1);
    V=V/M-Factor*Moy^2;
  Dev=1.96*sqrt(V)/sqrt(M);
  res=8*[Moy-Dev,Moy,Moy+Dev ];
endfunction
```